
Abstract

We formalize, in the Lean theorem prover, the Bruck-Ryser-Chowla theorem from
combinatorial design theory. Along the way, we formalize fundamental results about
balanced incomplete block designs (BIBDs). Many of these proofs rely on linear alge-
bra, as it turns out that combinatorial properties of BIBDs are closely tied to algebraic
properties of their incidence matrices. Consequently, we also formalized some relevant
results from matrix theory and quadratic forms.

1 Introduction

Formalization is more than verification—it is a process of reflection. It not only digitizes
mathematics but also invites us to see familiar structures in sharper resolution. In this
study, we embark on such a process for a classic topic in combinatorics: the theory of
combinatorial designs. Using the Lean theorem prover [3], we formalize a foundational body
of related results in design theory culminating with a proof of the Bruck-Ryser-Chowla
theorem, following the exposition of [5]. What might first appear as a technical exercise
is, in practice, an illuminating journey through the interplay between combinatorics, linear
algebra, and formal proof.

One might ask: Why formalize such familiar material? After all, the key theorems in
design theory have been known and carefully written down for decades. But writing down
mathematics for a human audience and encoding it for a proof assistant are not the same
activity. The latter demands more than correctness—it demands explicit structure. This gap
is not a flaw of the proof assistant but a feature: it reveals which ideas in our paper proofs
are robust and which rely, perhaps unknowingly, on informal reasoning or even ambiguities
in definitions. For example, the choice to represent a design not as a multiset of blocks, but
as a family of finite sets indexed by a type, is not a trivial one. It affects the ease of defining
incidence matrices, the applicability of counting arguments, and the feasibility of algebraic
generalizations.

The central objects of our formalization are balanced incomplete block designs (BIBDs),
which are combinatorial structures with deep algebraic fingerprints. The core idea—that
we can translate combinatorial constraints into algebraic identities of incidence matrices—
resonates well with the capacities of Lean. To capture this interaction, we not only develop
a hierarchy of design structures (including RPBDs and BBDs) but also build the linear

algebraic machinery to support their analysis. Many proof of combinatorial statements admit
algebraic proofs, especially through matrix manipulations. The prevalence of this style of
proof is a major reason for using Lean. Indeed, Lean’s mathematical library, mathlib [2],
already has a well developed collection of results from linear algebra and matrix theory.

Beyond the specific theorems we prove, this work is also a case study in designing formal
definitions that balance mathematical fidelity with proof assistant ergonomics. We highlight,
for instance, how our encoding choices facilitate the use of double counting arguments in
Lean, and how certain design axioms translate into precise matrix properties. We hope the
experience of formalizing this material not only supports future formal work in combinatorics,
but also brings a fresh perspective on the theory itself.

2 Designs and incidence matrices

A design is a finite set X together with a finite collection (multiset) B of subsets of X. The
elements of B are called blocks. In particular, a design may have repeated blocks.

Example 2.1. Take X = {0, 1, 2} and B = [{0, 1}, {1, 2}, {0}, {0}]. Then (X,B) is a design.

Example 2.2 (Fano plane). Take

X = {A,B,C,D,E, F,G}
B = [{A,F,B}, {B,D,C}, {C,E,A}, {A,G,D}, {B,G,E}, {C,G, F}, {D,E, F}]

The design (X,B) is called the Fano plane and can be drawn as follows, where the lines and
the circle represent the blocks.

A

B CD

EF
G

In Lean, we define a design as a structure parameterized by finite types X and ι, where
X corresponds to the underlying set of points, and ι is an index set. Finset X is the type
of finite sets whose elements have type X.

structure Design (ι X : Type*) [Fintype X] [Fintype ι] where

blocks : ι → Finset X

Example 2.3. To implement Example 2.1 using the formal definition, we may set X to Fin

3 (the finite type consisting of natural numbers less than 3), ι to be Fin 4, and define blocks
by

0 7→ {0, 1}, 1 7→ {1, 2}, 2 7→ {0}, 3 7→ {0}.

Remark 2.4. Note that our axiomatization of designs already differs from the literal trans-
lation of the textbook definition. Lean knows about multisets: the type Multiset α is
defined to be List α/ ∼, where List α is the type of lists on α, and ∼ is the “permutation
of” equivalence relation. A literal translation of the definition of a design would have been
to define blocks : Multiset (Finset X). There are plenty of reasons for our choice to
instead index the blocks by ι; one is that it makes incidence matrices very easy to define.

The incidence matrix of a design Φ = (X,B), withX = {x1, . . . , xv} and B = [B1, . . . , Bb]
is the v × b matrix M(Φ) where

M(Φ)i,j =

®
1 if xi ∈ Bj

0 else

As we will see later, certain combinatorial properties of designs translate into algebraic
conditions on the corresponding incidence matrices. In Lean, we have:

def toIncMat (α : Type*) [One α] [Zero α] (Φ : Design ι X) :

Matrix X ι α :=

of (fun x i 7→ if x ∈ Φ.blocks i then 1 else 0)

Here, Matrix X ι α is the type of matrices whose rows are indexed byX, columns are indexed
by ι, and elements have type α. The typeclasses [One α] and [Zero α] mean that α has
elements called “0” and “1”; although this is all the structure we need to define the incidence
matrix of a design, generally α will be a commutative ring or a field.

Example 2.5. The incidence matrix of the design described in Example 2.3 isÑ
1 0 1 1
1 1 0 0
0 1 0 0

é
.

For natural numbers v, b, l, a (v, b, l)-balanced block design (BBD) is a design (X,B) where
v = |X|, b = |B|, and which satisfies the balance property: for all x ̸= y ∈ X, the number of
blocks which contain both x and y is l.

structure BBD (ι X : Type*) [Fintype X] [DecidableEq X] [Fintype ι]
(l : N) extends Design ι X where

balance : ∀ x y, x ̸= y → #{i | x ∈ blocks i ∧ y ∈ blocks i} = l

A (v, b, l, r)-regular pairwise balanced design (RPBD) is a (v, b, l)-BBD (X,B) satisfying the
regularity property: for all x ∈ X, the number of blocks which contain x is r. A trivial
RPBD is an RPBD where every block has size v. Otherwise, it is a nontrivial RPBD.

structure RPBD (ι X : Type*) [Fintype X] [DecidableEq X] [Fintype ι]
(l r : N) extends BBD ι X l where

regularity : ∀ x, #{i | x ∈ blocks i} = r

structure nontrivialRPBD (ι X : Type*) [Fintype X] [DecidableEq X] [Fintype ι]
(l r : N) extends RPBD ι X l r where

nontrivial : ∃ i, 0 < #(blocks i) ∧ #(blocks i) < Fintype.card X

Finally, a (v, b, k, l)-balanced incomplete block design (BIBD) is a (v, b, l)-BBD such that
v > k ≥ 2 and each block has size k.

structure BIBD (ι X : Type*) [Fintype X] [DecidableEq X] [Fintype ι]
(k l : N) extends BBD ι X l where

hvk : k < Fintype.card X ∧ 2 ≤ k

hA : ∀ i : ι, #(blocks i) = k

Note that Example 2.2, the Fano plane, is a (7, 7, 3, 1)-BIBD.
Although the majority of our formalization is about BIBDs, this hierarchy of structures

emphasizes which properties are actually being used in certain algebraic characterizations of
designs.

3 A counting principle

A fundamental result of BIBDs is that any BIBD is also an RPBD.

Theorem 3.1. In a (v, b, k, l)-BIBD (X,B), every x ∈ X occurs in l(v − 1)/(k − 1) blocks.

Proof. Fix x ∈ X, and let r be the number of blocks in which x occurs. We count

S = {(y,B) | x ̸= y, x ∈ B, and y ∈ B} ⊆ X × B

in two ways. On one hand, there are v − 1 ways to choose y distinct from x, and for each
of those choices of y, there are l ways to choose a block B containing both x and y by the
balance property. So #S = l(v − 1). On the other hand, there are r ways to choose B
containing x, and for each such choice, there are k− 1 ways to choose x ∈ B distinct from y.
So #S = r(k−1). Equating these two expressions for #S, we obtain l(v−1) = r(k−1).

The value r is called the replication number of the BIBD, and we may use the notation
(v, b, k, l, r)-BIBD to spell out each of the parameters.

Given Φ : BIBD ι X k l, we define the replication of an element x : X via

def rep_elem (Φ : BIBD ι X k l) (x : X) :=

#{i | x ∈ Φ.blocks i}

We then prove that

theorem rep_constant : ∀ x, (k - 1) * rep_elem Φ x = l * (v - 1) := by...

which in particular implies that the value of rep_elem Φ x is independent of x, as long as X
is inhabited. This allows us to define the replication number of Φ to be the common value
of the replication of each element.

def rep [Inhabited X] (Φ : BIBD ι X k l) : N :=

rep_elem Φ default

In the above definition, default provides the “default” element of the inhabited type X.
Similar double counting arguments establish the following theorem.

Theorem 3.2. Any (v, b, k, l, r)-BIBD satisfies bk = vr.

Proof. We count
S = {(x,B) | x ∈ B} ⊆ X × B

in two ways. On one hand, there are v ways to first choose x, then r ways to choose B such
that x ∈ B; so #S = vr. On the other hand, there are b ways to first choose B, then k ways
to choose x such that x ∈ B; so #S = bk. This establishes the equality.

The key ingredient in the proof of Theorem 3.1 and Theorem 3.2 is the following counting
principle.

Theorem 3.3 (Counting principle). Let α, β be finite sets and P,Q predicates on α and
α× β respectively. Suppose it is known that there are n choices of x such that P (x) holds,
and for each x ∈ α satisfying P (x), there are k choices of y such that Q(x, y) holds. Then

#{(x, y) | P (x) ∧Q(x, y)} = kn.

Proof. Although the statement is intuitively quite clear, work has to be done to explain the
proof to a computer, or to a stubborn human. Our formalized proof proceeds along the
following lines. The given condition supplies us with bijections

gx : {y | Q(x, y)} ∼= {0, 1, . . . , k − 1}

for each x satisfying P (x). We may then define

f : {(x, y) | P (x) ∧Q(x, y)} → {x | P (x)} × {0, 1, . . . , k − 1}
(x, y) 7→ (x, gx(y)),

and it can be shown that

f−1 : {x | P (x)} × {0, 1, . . . , k − 1} → {(x, y) | P (x) ∧Q(x, y)}
(x, j) 7→ (x, g−1

x (j)).

is the inverse of f , and therefore f is a bijection.

The statement of Theorem 3.3 in Lean is as follows:

theorem card_dependent {α β : Type*} [Fintype α] [Fintype β]
(P : α → Prop) [DecidablePred P]

(Q : α → β → Prop) [∀ x, DecidablePred (Q x)]

{k : N} (hk : ∀ x, P x → #{y | Q x y} = k) :

#{(x, y) | P x ∧ Q x y} = k * #{x | P x} := by...

As an example, let’s apply Theorem 3.3 to Theorem 3.1. Let Φ : BIBD ι X k l and fix
x an element of X. As before, we define

S = {(y, i) | x ̸= y ∧ x ∈ Φ.blocks i ∧ y ∈ Φ.blocks i}.

Then apply the Theorem 3.3 with the predicates P : X → Prop and Q : X × ι → Prop
defined by

P : y 7→ (x ̸= y), and

Q : (y, i) 7→ (x ∈ Φ.blocks i ∧ y ∈ Φ.blocks i)

to obtain #S = (#X − 1)× l. Likewise, apply Theorem 3.3 again with P ′ : X → Prop and
Q′ : X × ι → Prop defined by

P ′ : x 7→ (x ∈ Φ.blocks i), and

Q′ : (y, i) 7→ (x ̸= y ∧ y ∈ Φ.blocks i)

to obtain #S = r × (k − 1), where as usual r is the replication number of Φ.

Remark 3.4. In Remark 2.4, we explained one reason for preferring blocks : ι → Finset X

over blocks : Multiset (Finset X). Another reason is that, had we chosen the latter ax-
iomatization, it is not possible to directly apply the counting principle to prove Theorem 3.1
and Theorem 3.2. In fact, it is not clear to even state an analogous counting principle
adapted for Multiset.

4 Algebraic properties of designs

Results about designs tend to rely on algebraic properties of their associated incidence ma-
trices. The proofs we present in this section were chosen for the ease with which they can
be formalized; indeed, they are direct transcriptions of the formalized proofs.

Any v × b zero-one matrix M gives rise to a design (X,B) where X = {0, 1, . . . , v − 1}
and B = [B0, B1, . . . , Bb−1] where Bj = {i ∈ X | Mi,j = 1}. In Lean, we have:

def fromIncMat (α : Type*) [DecidableEq α] [One α] (M : Matrix X ι α) :

Design ι X where

blocks := fun i 7→ {x | M x i = 1}

Crucially, we have the following algebraic characterization of RPBDs.

Theorem 4.1. A v × b zero-one matrix M is the incidence matrix of a (v, b, l, r)-RPBD if
and only if MMT = lJv + (r − l)Iv, where Jv is the all-ones matrix and Iv is the identity.

Proof. Suppose M is the incidence matrix of a (v, b, l, r)-RPBD, and let Mi denote the ith
row of M . Then (MMT )i,j is the dot product ⟨Mi,Mj⟩, which is precisely the number of
blocks which contain both xi and xj. If i ̸= j, this number is l by the balance property;
otherwise, this number is r by Theorem 3.1.

In the other direction, suppose M is a 0 − 1 matrix satisfying MMT = lJv + (r − l)Iv.
Let Φ = (X,B) be the design associated with M . We claim that Φ is an RPBD. Indeed, for
each xi ∈ X, we have

#{B ∈ B | xi ∈ B} = ⟨Mi,Mi⟩ = (MMT )i,i = r,

and likewise for each xi ̸= xj ∈ X, we have

#{B ∈ B | xi ∈ B and xj ∈ B} = ⟨Mi,Mj⟩ = (MMT )i,j = l,

so Φ satisfies regularity and the balance property.

This result is enough to deduce Fisher’s Inequality.

Theorem 4.2 (Fisher’s Inequality). Any nontrivial (v, b, l, r)-RPBD Φ satisfies v ≤ b.

Proof. Let M = M(Φ) be the incidence matrix of Φ. By the matrix determinant lemma,
Theorem 4.1, and the rank inequality, we have

v = rank(lJv + (r − l)Iv) = rank(MMT ) ≤ rank(M).

Since M is a v × b matrix, this implies v ≤ b.

The rank inequality is in Mathlib, but the matrix determinant lemma is not. It states
that for a n× 1 matrix u and a 1× n matrix v,

det(I + uv) = 1 + vu.

theorem det_one_add_column_mul_row {α n : Type*}

[CommRing α] [Fintype n] [DecidableEq n]

(u : Matrix n (Fin 1) α) (v : Matrix (Fin 1) n α) :

det (1 + u * v) = 1 + (v * u) 0 0 := by...

In Lean, the proof is a straightforward manipulation of block matrix multiplication. From
this we obtain a formula for det(lJv + (r − l)Iv) and show it is nonzero.

A symmetric BIBD is a (v, b, k, l, r)-BIBD Φ with v = b (equivalently k = r, by Theo-
rem 3.2). For example, the Fano plane (Example 2.2) is a symmetric BIBD. The incidence
matrix M(Φ) of a symmetric BIBD Φ is square (but not necessarily symmetric!). Moreover,
by inspecting the proof of Theorem 4.2, one sees that M(Φ) has full rank.

Theorem 4.3. Let Φ be a symmetric (v, b, k, l, r)-BIBD. Then for any pair of distinct blocks
Bi and Bj, we have #(Bi ∩Bj) = l.

Proof. Let M = M(Φ). By definition, for all i ̸= j, (MTM)i,j = #(Bi ∩ Bj). Moreover,
(MJ)i,j = r = k = (JM)i,j for all i, j, so MJ = JM . By Theorem 4.1,

MMTM = (lJv + (r − l)Iv)M = M(lJv + (r − l)Iv).

Since M is invertible, it follows that

MTM = lJv + (r − l)Iv

which implies the conclusion.

In much the same style of proof, we also have:

Theorem 4.4. Any (v, b, l, r)-RPBD with v = b is a BIBD.

Proof. Let Φ be a (v, b, l, r)-RPBD with incidence matrix M . It suffices to show that every
block Bi of Φ has the same size. Let 1 denote the all-ones vector of dimension v. By
definition,

(MT1)i = #Bi and M1 = r1.

By Theorem 4.1,

MMT1 = (lv + r − l)1 = M

Å
lv + r − l

r
1

ã
,

so invertibility of M implies that

MT1 =
lv + r − l

r
1,

which implies that every block of Φ has size (lv + r − l)/r.

5 Existence of symmetric BIBDs

A central problem of combinatorial design theory seeks necessary conditions for the existence
of symmetric BIBDs. Currently, the Bruck-Ryser-Chowla theorem gives the strongest known
necessary condition. The theorem gives conditions for the cases of v even and v odd. The
tools we have developed so far are enough to easily deduce the even case.

Theorem 5.1 (Bruck-Ryser-Chowla, v even). Let v be even, and let Φ be a (v, v, k, l, r)-
BIBD. Then k − l is a perfect square.

Proof. Compute

det(M)2 = det(MMT ) = det(lJv + (r − l)Iv)

= (k + (v − 1)l)(k − l)v−1 = k2(k − l)v−1,

where we use the matrix determinant lemma and Theorem 3.1. Since detM is an integer,
the result follows.

The odd case has more interesting proof and formalization. Before stating it, we discuss
the ingredients of the proof.

Let F be a field and M ∈ Fn×n. Then M determines the quadratic form

Fn ∋ x 7→ ⟨x,Mx⟩ ∈ F.

We say that the quadratic forms over F on n variables p and q are equivalent if their ranges
coincide, i.e.

{p(x) | x ∈ Fn} = {q(x) | x ∈ Fn}.

Given matrices M,N ∈ Fn×n, we say that M is congruent to N , written M ∼ N , if there
is some invertible matrix Q ∈ Fn×n such that M = QNQT . Note that ∼ is an equivalence
relation on Fn×n. It is easy to show that M ∼ N just if their associated quadratic forms are
equivalent. In Lean, we have:

structure MatCongr {n α : Type*} [Fintype n] [DecidableEq n] [CommRing α]
(M N : Matrix n n α) where

Q : Matrix n n α
inv : Invertible Q

cong : M = Q * N * Q⊤

infixl:55 " ∼m " => MatCongr

Theorem 5.2. Let n be a multiple of 4 and let F be a field of characteristic 0. Then for
any positive integer m, mIn ∼ In, where In ∈ Fn×n is the identity.

Proof sketch. Follows from the Legendre four squares theorem.

noncomputable def matCongrOneOfFourDiv [CharZero α]
(hn : 4 | Fintype.card n) {m : Z} (mpos : 0 < m) :

(m : α) • (1 : Matrix n n α) ∼m (1 : Matrix n n α) := by...

The direct sum of M ∈ Fm×m and N ∈ Fn×n, written M ⊕N , is the block matrix

M ⊕N =

Å
M 0
0 N

ã
∈ F(m+n)×(m+n).

def matDirectSum (M : Matrix m m α) (N : Matrix n n α) :=

fromBlocks M 0 0 N

infixl:60 " ⊕m " => matDirectSum

The key result used by the proof of Theorem 5.4 is the so-called Witt cancellation lemma,
which shows how direct sums interact with congruence.

Theorem 5.3 (Witt Cancellation). Let A,B,C be symmetric matrices over a field in which
2 ̸= 0. Suppose C ⊕ A ∼ C ⊕B. Then A ∼ B.

Proof. See [1].

The tools are now in place to prove the harder case of the Bruck-Ryser-Chowla theorem.
We present only a sketch of the proof given in [4] to give a sense of the kind of calculation
we formalized.

Theorem 5.4 (Bruck-Ryser-Chowla, v odd). Let v be odd, and let Φ be a (v, v, k, l, r)-BIBD.
Then there exists a nontrivial integer solution (x, y, z) to the equation

x2 = (k − l)y2 + (−1)
v+1
2 lz2.

Proof sketch. Suppose v ≡ 1 (mod 4). All matrices in the following calculation have entries
in Q:

Iv−1 ⊕ I1 ⊕−lI1 ∼ Iv ⊕−lI1 (Iv−1 ⊕ I1 = Iv)

∼ (k − l)Iv ⊕−k − l

l
I1 (Theorem 4.1 plus some work)

∼ (k − l)Iv−1 ⊕ (k − l)I1 ⊕−k − l

l
I1 (Iv−1 ⊕ I1 = Iv)

∼ Iv−1 ⊕ (k − l)I1 ⊕−k − l

l
I1. (Theorem 5.2)

Then Theorem 5.3 allows us to cancel the Iv−1 from both sides, leavingÅ
1 0
0 −l

ã
∼
Å
k − l 0
0 −k−l

l

ã
.

Thus the quadratic forms

p(x, y) = x2 − ly2 and q(x, y) = (k − l)x2 − k − l

l
y2

are equivalent over Q, which implies the conclusion. The case that v ≡ 3 (mod 4) proceeds
along similar lines.

The meat of the proof is the calculation involving direct sums and congruence. The main
difficulty and difference between the ordinary language proof and the formalization is that,
on paper, we wouldn’t hesitate to write something like

“M ⊕ (N ⊕ P ) ∼ (M ⊕N)⊕ P”.

Now, in Lean, suppose M : Matrix m m α, N : Matrix n n α, and P : Matrix p p α. Then
the above expression is not well-typed because the types m ⊕ (n ⊕ p) and (m ⊕ n) ⊕ p are
not definitionally equal (even though they are isomorphic). Therefore, to implement the
calculation in the proof of Theorem 5.4, we had to be very careful to make sure the types
matched the whole way through. We could reassociate parentheses with the help of the
following lemma:

def matCongrAssocOfMatCongr

{M M’ : Matrix m m α} {N N’ : Matrix n n α} {O O’ : Matrix o o α}
(h : M ⊕m N ⊕m O ∼m M’ ⊕m N’ ⊕m O’) :

M ⊕m (N ⊕m O) ∼m M’ ⊕m (N’ ⊕m O’) := by...

Remark 5.5. It might be tempting to try and define matrix congruence between matrices
of different index types, given that they are isomorphic:

structure MatCongr’ {m n α : Type*} [Fintype m] [DecidableEq m] [CommRing α]
(N : Matrix n n α) (M : Matrix m m α) extends n ≃ m where

A : Matrix m m α
inv : Invertible A

cong : reindex toEquiv toEquiv N = A * M * A⊤

This definition seems very natural and much more flexible than the one we adopt. Unfor-
tunately, it fails to capture the notion of matrix congruence! The problem is that the given
isomorphism e : n ≃ m might not be the one we expect. In general, it is not true that
M ∼ PM for all matrices M and permutation matrices P . For instance,Å

1 0
0 1

ã
and

Å
0 1
1 0

ã
are not congruent because their associated quadratic forms are x2+ y2 and 2xy respectively,
which clearly are not equivalent.

